
付録　グラフ表示を含むモデル

1 はじめに

　本稿では，進化ゲーム（Evolutionary Game）のシミュレーションモデルをCAST言語を用いて構成し実行する．具体的には，競合関係が三すくみの関係にある3種類の生物の人口分布の変化を動的に計算する[注3]．三すくみの関係とは，生物1は生物2より強く，生物2は生物3より強く，生物3は生物1より強いことを言う．（じゃんけんでの，グー，チョキ，パーの関係に相当する．）ここでは，人口分布のバランスが崩れた場合に，その後の人口分布がどのように変化してゆくかをシミュレートする．例えば一時的に生物1（いつでもグーだけ出す人）の人口が増加したと仮定しよう．しかしながら，餌となる生物3（チョキ）が少なくなっていることに注意しなければならない．餌に出会えない生物1の子孫は減少していくかもしれない．また，生物3（パー）は生物1（グー）より強いので，確実に餌にありつけて子孫を残すことができるだろう．時間が経過すると，最終的にはバランスのとれた人口分布に戻るだろう．このバランスのとれた状態を（多型集団の）進化的安定状態（Evolutionarily Stable State: ESS）と呼ぶ．

　このサイト「モデル記述言語CASTによるシミュレーション」[注1]で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境Simcastができあがり，本文中のモデルを実行することができる．開発実行環境Simcastの操作方法などは，同サイトや教科書[注2]を参照されたい．

2  進化ゲーム
　限られた領域内で餌をめぐって競合している３種の生物1,2,3が混在する生態系（多型集団）を考える．全個体数は無限とし，現在の人口分布を c = (c1, c2, c3)とする．ただし，c1+c2+c3=1である．各個体は１対１ランダム対戦を行ない，適応度w1, w2,w3が定まることとする．次世代の人口分布cc = (cc1, cc2, cc3)は，各生物の適応度が高い方がより多く子孫を残せるように決まるものとする（図表1）

現世代の人口分布 c = (c1,c2,c3)

↓

ランダム対戦

↓

適応度 w = (w1,w2,w3)

↓

次世代の人口分布 cc = (cc1,cc2,cc3)

図表1　人口分布の変化

　具体的には次のように定式化する[注3]．
　　　w1 = 5*c2 -4*c3 +20, w2 = -7*c1 +8*c3 +20, w3 = 2*c1 -c2 +20,  --- (1)

　　　w = c1*w1+c2*w2+c3*w3,                                 --- (2)
　　　cc1 = c1*w1/w, cc2 = c2*w2/w, cc3 = c3*w3/w .                --- (3)
　式(1)における値20は生得的な適応度である．それらにランダム対戦後の餌の獲得量を加算している．係数の正負から競合関係が三すくみの関係にあることも分かる．式(2)は適応度の平均値wを求めるものであり，式(3)は適応度が高いほど次世代に多くの子孫を残せることを表している． 

　さて，このような計算と描画を行なうために，モデル理論アプローチではオートマトンを作成する．オートマトン（正確には状態機械）の状態は人口分布とする．c = (c1,c2,c3)．一回の状態遷移で1世代が経過するものとする．ここでは1000世代の変化を見たいので，オートマトンは1000回の状態遷移を行なうことになる．

3　言語CASTによる表現
　上記の計算を行なうオートマトンのCAST言語によるモデルは次のように構成できる．


//evogame1.set

inputsequence()="d"; times()=3000;

initialstate()=c0 <-> c0:=[0.5,0.3,0.2];

delta([c1,c2,c3],a)=cc <-> 

  c:=[c1,c2,c3], 

  d1:=[0,5,-4], d2:=[-7,0,8], d3:=[2,-1,0],

  [w1,w2,w3]:=sum([d1*c,d2*c,d3*c])+20, 

  w:=c1*w1+c2*w2+c3*w3, r:=1,

  [cc1,cc2]:=[c1+(r*c1*(w1-w)/w),c2+(r*c2*(w2-w)/w)],

  cc:=[cc1,cc2,1-cc1-cc2];

　　　　図表2　進化ゲームのモデル

　図表2のモデルでは，まず初期状態がc0:=[0.5,0.3,0.2]から始まる．状態遷移関数deltaは，前述の式(1), (2), (3)を計算している．また，強制的に時間を進めるためにダミーの"d"を3000回入力している（inputsequence()="d"; times()=3000;）．図表3はSimcastでコンパイルし実行した結果である．3000回の状態遷移を行ない，最終的にc = (0.31142, 0.34165, 0.34695)が得られている．

図表3　実行結果
　図表2のモデルでは，式(3)に無かった変数rがある．cc1の定義式を展開すれば分かるように，r=1のときは式(3)と同じであり，離散時間モデルを表わしている．ここで，r=0.01などに変更した場合は，連続時間モデル（微分方程式を離散化したもの）になる．その結果，数値は少しずつ変化するようになるだろう．

4　グラフ表示

　以上が計算の本質的なところであるが，グラフを表示するために，若干の述語を追加したものを付録に掲載した．付録のモデルevogame5.setをSimcastでコンパイルし実行すると，図表4のように3角形の中に渦巻き状の図形が描かれていく．

図表4　グラフ表示の結果

　図表4の3角形の

　　下左端は生物1が多い状態 c = (1,0,0)，

　　下右端は生物2が多い状態 c = (0,1,0)，

　　上端は生物3が多い状態　 c = (0,0,1)

に対応している．描画の結果を見ると，生物1が多い初期状態から出発し次第に生物3が増加して行く．しかし，ある程度生物3が増加すると，こんどは生物2が増加していく様子が分かる．時間が経過するにつれ，外側から内側に渦巻き状に点が描かれて行き，最終的には3角形の中心，つまり人口分布が(1/3,1/3,1/3)に近づいていくことが分かる．これは，c=(1/3,1/3,1/3)が進化的安定状態であることを意味する．
　CAST言語による付録のモデルevogame4.setでは，まず，事前準備preprocessとして起動直後にスプレッドシートとグラフウインドウを開いている（述語makesheetとwopen）．またplaingraph(Wp2.g)はグラフウインドウの座標軸を消去する述語である．さらに，関数defListを使って外枠の3角形のグラフを描いている．ただし，変数uはダミーであり常に空である．

　スプレッドシートは現在の状態を数値で表示するためのものである．時間の経過とともに状態c = (c1,c2,c3) が変化していく様子が表示される．しかし，3000回の状態遷移が終了したあと，再びevogame5.setを実行すると「続きの3000回」を実行することができるように，初期状態をスプレッドシートから読み込むように定義している（initialstate()=c0 <-> [c0]:=projectss(Wp.g,1,2,1,4)）．
　グラフウインドウは，状態cの遷移を図形として描くためのものである．述語showPoint(c)を作成し，3角形の中に点を描くことができる．変数c=(c1,c2,c3)は3次元空間の点であるが，c1+c2+c3=1であるので，実際には空間座標(1,0,0), (0,1,0), (0,0,1)を端点とする3角形（空間上の平面）上の点である．したがって，showPoint(c)は，点c=(c1,c2,c3)をグラフウインドウ（2次元）上に描くための座標変換（ (c1,c2,c3)→(x,y) ）を行っている．
5 おわりに

　本稿では，進化ゲームのモデルをCAST言語を用いて構成し，モデル理論アプローチによるシミュレーション開発実行環境Simcastで実際に実行し，状態遷移の経過を描画した．その結果，それなりの計算はできるし一応のグラフ表示が可能であることが分かった．

　しかしながら，外枠の3角形を描くのに複数の点で表示したが，直線を描く適切な述語が必要である．し，不要な座標軸を消せないことも課題として残っている．

注

付録（グラフ表示を含むモデル）
//evogame5.set

preprocess() <-> 

  getWp("sheet",Wp), (Wp>0)->(Wp.g:=Wp) 

    otherwise ( makesheet(2,5,Wp1), Wp.g:=Wp1,

    writess(Wp.g,"r",1,1,[" State",0.5,0.3,0.2]) ),

  getWp("graph",Wp2), (Wp2>0)->(Wp2.g:=Wp2) 

    otherwise ( wopen(Wp3,"xyplot"), plaingraph(Wp3), Wp2.g:=Wp3 ),

  u:=defList(p(y,x,[]),member(x,genIndex(0,50)));

p(y,x,[]) <-> 

  showPoint(2,[1-x/50,x/50,0]),  //3角形を描く

  showPoint(2,[0,1-x/50,x/50]),

  showPoint(2,[x/50,0,1-x/50]);
inputsequence()="d"; times()=1000;

initialstate()=c0 <-> [c0]:=projectss(Wp.g,1,2,1,4);

delta([c1,c2,c3],a)=cc <-> 

  c:=[c1,c2,c3], d1:=[0,5,-4],d2:=[-7,0,8],d3:=[2,-1,0],

  [w1,w2,w3]:=sum([d1*c,d2*c,d3*c])+20, 

  w:=c1*w1+c2*w2+c3*w3, r:=0.1, 

  [cc1,cc2]:=[c1+(r*c1*(w1-w)/w),c2+(r*c2*(w2-w)/w)],

  cc:=[cc1,cc2,1-cc1-cc2];

lambda(c)=b <-> b:=0, 

  showPoint(0,c),

  writess(Wp.g,"r",1,2,c);

showPoint(z,[c1,c2,c3]) <->

  aX:=30, aY:=220, length:=240,

  x:=aX +((1-c1+c2)*length/2),

  y:=aY -(c3*sqrt(3)*length/2),

  xwrite(Wp2.g,"r",[x,y,x+z,y+z]);

